7

Points, lines and planes

7.1 Distance Between Two Points in Space



Given two points in space (x_K, y_K, z_K) and (x_L, y_L, z_L) the distance between them, r, is given by Pythagoras theorem as in the two dimensional case described in Section 1.1:

$$r = \sqrt{[(x_L - x_K)^2 + (y_L - y_K)^2 + (z_L - z_K)^2]}$$

All the computational considerations mentioned in Section 1.1 apply in the three-dimensional case as well. In particular it may often be more convenient and quicker to encode an algorithm using squared distances throughout a data structure than to waste time using the expensive square root function.

7.2 Equations of a Straight Line in Space

In three dimensions, the implicit equation of a line is the intersection of two planes. Since more than one pair of planes can describe a given line, the parametric form is to be preferred. This is very similar to the two-dimensional parametric equation of a line:

$$x = x_0 + ft$$

$$y = y_0 + gt$$

$$z = z_0 + ht$$

This may be specified so that the parameter, t, has values 0 and 1 at the ends of a segment (see Section 7.6). Alternatively, the normalised form may be prefered, and, as in two dimensions, this means that a change in t corresponds to real distance moved along the line. When the equations are normalised the condition

$$f^2 + g^2 + h^2 = 1$$

must be satisfied. This can be achieved by dividing f, g, and h by $\sqrt{(f^2 + g^2 + h^2)}$ to obtain new values:

```
DENSQ = F*F + G*G + H*H
IF (DENSQ.LT.ACCY) THEN

        ..... The parametric coefficients are corrupted

ELSE
        DINV = 1.0/SQRT(DENSQ)
        F = F*DINV
        G = G*DINV
        H = H*DINV
ENDIF
```

In the normalised form the coefficients f, g, and h are the cosines of the angles the line makes with the coordinate axes.

The remaining arbitrariness in these equations can be removed by making the point (x_0, y_0, z_0) the point on the line nearest to the origin of coordinates. This is the point where the normal from the origin meets the line. This may avoid numerical problems with points (x_0, y_0, z_0) that are very distant from the origin, and it may facilitate comparisons of several lines, especially to detect coincident lines. Note that an accuracy constant must be used when comparing floating point numbers.

The normal from from the origin to a parametric line meets the line where

$$t = \frac{-(fx_0 + gy_0 + hz_0)}{(f^2 + g^2 + h^2)}$$

but when the line is in its normalised form the bottom line may be omitted, as it is 1. The t = 0 point of a normalised line may be moved to the point nearest the origin as follows:

```
D = F*XO + G*YO + H*ZO
XO = XO - F*D
YO = YO - G*D
ZO = ZO - H*D
```

7.3 Distance from a Point to a Line in Space



If the line equation is

$$x = x_0 + ft$$

$$y = y_0 + gt$$

$$z = z_0 + ht$$

then the value of the parameter, t, at the point on the line nearest to the given point (x_J, y_J, z_J) is:

$$t = \frac{f(x_J - x_0) + g(y_J - y_0) + h(z_J - z_0)}{(f^2 + g^2 + h^2)}$$

If the line equation is normalised the denominator is 1 of course. If we let $x_{J0} = (x_J - x_0)$, and so on, then the squared distance from the point to the line, r^2, is given by:

$$r^2 = \frac{[g(fy_{J0} - gx_{J0}) + h(fz_{J0} - hx_{J0})]^2 + [f(gx_{J0} - fy_{J0}) + h(gz_{J0} - hy_{J0})]^2 + [f(hx_{J0} - fz_{J0}) + g(hy_{J0} - gz_{J0})]^2}{(f^2 + g^2 + h^2)^2}$$

This can be coded:

```
DENOM = F*F + G*G + H*H
IF (DENOM.LT.ACCY) THEN

     ..... The line parameter coefficients are corrupt

ELSE
     XJO = XJ - XO
     YJO = YJ - YO
     ZJO = ZJ - ZO

     FYGX = F*YJO - G*XJO
     FZHX = F*ZJO - H*XJO
     GZHY = G*ZJO - H*YJO

     V1 = G*FYGX + H*FZHX
     V2 = H*GZHY - F*FYGX
     V3 = - F*FZHX - G*GZHY

     R = SQRT(V1*V1 + V2*V2 + V3*V3)/DENOM
ENDIF
```

7.4 Distance Between Two Lines in Space



If the two lines have parametric equations

$$x = x_1 + f_1 s$$

$$y = y_1 + g_1 s$$

$$z = z_1 + h_1 s$$

and

$$x = x_2 + f_2 t$$

$$y = y_2 + g_2 t$$

$$z = z_2 + h_2 t$$

then the minimum distance between them is given by the expression:

$$r = \frac{\begin{vmatrix} x_2 - x_1 & y_2 - y_1 & z_2 - z_1 \\ f_1 & g_1 & h_1 \\ f_2 & g_2 & h_2 \end{vmatrix}}{\sqrt{\begin{vmatrix} f_1 & g_1 \\ f_2 & g_2 \end{vmatrix}^2 + \begin{vmatrix} g_1 & h_1 \\ g_2 & h_2 \end{vmatrix}^2 + \begin{vmatrix} h_1 & f_1 \\ h_2 & f_2 \end{vmatrix}^2}}$$

Considerable exploitation of subexpressions may be made in coding this, which then becomes:

```
X21 = X2 - X1
```

```
Y21 = Y2 - Y1
Z21 = Z2 - Z1

FG = F1*G2 - F2*G1
GH = G1*H2 - G2*H1
HF = H1*F2 - H2*F1

DENOM = FG*FG + GH*GH + HF*HF
IF (DENOM.LT.ACCY) THEN

        ..... The lines are parallel

ELSE
        R = ABS(X21*GH + Y21*HF + Z21*FG)/SQRT(DENOM)
ENDIF
```

The sign of the expression is an indication of the relative parametric direction of the two lines. This information is much more easily obtained in other ways (see Section 7.5) and so the sign is discarded.

Note that this code does not work for parallel and near-parallel lines. The geometry in such cases is inherently unstable, and it is best to reformulate the problem itself as one of finding the distance from a *point* to a line (see Section 7.3).

7.5 Angle Between Two Lines in Space



The angle between two lines is found from the scalar product of their direction vectors; it is not necessary for the lines to intersect. For the two lines

$$x = x_1 + f_1 s$$

$$y = y_1 + g_1 s$$

$$z = z_1 + h_1 s$$

and

$$x = x_2 + f_2 t$$

$$y = y_2 + g_2 t$$

$$z = z_2 + h_2 t$$

with normalised coefficients, the angle between them is given by:

$$\theta = \cos^{-1}(f_1 f_2 + g_1 g_2 + h_1 h_2)$$

If the lines are not normalised, then using the expression

$$\theta = \cos^{-1} \frac{f_1 f_2 + g_1 g_2 + h_1 h_2}{\sqrt{[(f_1^2 + g_1^2 + h_1^2)(f_2^2 + g_2^2 + h_2^2)]}}$$

avoids one square root operation and several divisions when compared with normalising the lines separately before finding the angle between them. This is coded:

```
DENOM = (F1*F1 + G1*G1 + H1*H1)*(F2*F2 + G2*G2 + H2*H2)
IF (DENOM.LT.ACCY) THEN

        ..... One or both lines have corrupted coefficients

ELSE
        THETA = ACOS((F1*F2 + G1*G2 + H1*H2)/SQRT(DENOM))
ENDIF
```

The ACOS function returns values in the range 0 to π.



The angle measured is that between the lines with parameters moving in the same direction. θ = 0 indicates that the lines are parallel in the same direction, and $\theta = \pi$ that they are parallel in the opposite direction. Note that rounding error makes it unlikely that results for parallel lines will give exactly 0 or exactly π.

If the acute angle is required values over $\pi/2$ must be subtracted from π.

For old FORTRAN compilers that do not have an ACOS function, an alternative using ATAN2 is described in Section 1.4.

7.6 Line through Two Points in Space



The specification of a parametric line in space given two points on it is a direct extension of the two-dimensional formula given in Section 1.6. Again the parameter value, t, goes from 0 at the first point to 1 at the second. The line equations are:

$$x = x_K + (x_L - x_K)t$$

$$y = y_K + (y_L - y_K)t$$

$$z = z_K + (z_L - z_K)t$$

This is not normalised unless the distance between the points is 1. It can be normalised by dividing the coefficients of t (but *not* the constant terms) by the square root of sum of squares of those coefficients.

7.7 Equation of a Plane

Both the implicit and the parametric equations of planes in space are useful. The implicit form is simpler in general, and more compact. The parametric form is to be preferred when essentially two-dimensional operations are to be performed *in* the plane, because it readily permits axes to be defined in it.

The Implicit Form

This is the direct equivalent of the implicit line equation in two dimensions, and is:

$$ax + by + cz + d = 0$$

In the normalised form $a^2 + b^2 + c^2 = 1$. To convert an unnormalised implicit plane equation to its normalised form multiply the entire equation by:

$$\frac{1}{\sqrt{(a^2 + b^2 + c^2)}}$$

In the normalised form a, b, and c are the cosines of the angles which the normal to the plane makes with the coordinate axes. The absolute value of d is the perpendicular distance of the origin from the plane. The distance between two parallel normalised planes, 1 and 2, may therefore be determined as $|d_2 - d_1|$.

As with the implicit line equation, a normalised plane equation may be multiplied through by −1. The plane may be considered to represent the boundary of a semi-infinite region of space by applying a convention that the vector formed by the direction cosines (a, b, and c) always points towards the outside (or the inside) of the region. With such a convention, the equation is that of a planar half-space, a boundary between a volume of solid and an empty volume. Such half-spaces can be used to define convex polyhedra and more complicated three-dimensional shapes and regions in space by using the set theoretic methods described in Section 4.5 for two dimensions.

The Parametric Form

A single parameter may be used to specify distance along a line. Two are required to specify a position in a plane. Given a point in space (x_0, y_0, z_0), and two *different* vectors (i.e. directions) both parallel to the plane (f_1, g_1, h_1) and (f_2, g_2, h_2), a point in the plane is found by adding a proportion of one vector and a different proportion of the second vector to the point coordinates:

$$x = x_0 + f_1 s + f_2 t$$

$$y = y_0 + g_1 s + g_2 t$$

$$z = z_0 + h_1 s + h_2 t$$

If the two vectors have unit length, and are perpendicular, that is the conditions

$$f_1^2 + g_1^2 + h_1^2 = 1$$

$$f_2^2 + g_2^2 + h_2^2 = 1$$

$$f_1 f_2 + g_1 g_2 + h_1 h_2 = 0 \quad \text{(scalar product = zero)}$$

are fulfilled, then the parameters s and t constitute measurements along orthogonal axes in the plane from an origin in it (x_0, y_0, z_0), and two-dimensional geometry of the type described in the first five chapters of this book may be performed in the plane by using s and t in place of x and y. We have, in fact, already met this in the section on true perspective projection. The final stages of the projection were two-dimensional geometry in an image plane perpendicular to the direction in which an eye was pointing (see Section 6.4).

Conversion Between Implicit and Parametric Planes

As with the parametric line, the point on the plane nearest to the origin of coordinates is often a convenient one to use as the origin of the parametric system. It corresponds to the parameter values

$$s = \frac{(f_2 x_0 + g_2 y_0 + h_2 z_0)(f_1 f_2 + g_1 g_2 + h_1 h_2) - (f_1 x_0 + g_1 y_0 + h_1 z_0)(f_2^2 + g_2^2 + h_2^2)}{(f_1^2 + g_1^2 + h_1^2)(f_2^2 + g_2^2 + h_2^2) - (f_1 f_2 + g_1 g_2 + h_1 h_2)^2}$$

$$t = \frac{(f_1 x_0 + g_1 y_0 + h_1 z_0)(f_1 f_2 + g_1 g_2 + h_1 h_2) - (f_2 x_0 + g_2 y_0 + h_2 z_0)(f_1^2 + g_1^2 + h_1^2)}{(f_1^2 + g_1^2 + h_1^2)(f_2^2 + g_2^2 + h_2^2) - (f_1 f_2 + g_1 g_2 + h_1 h_2)^2}$$

which can, of course, be much simplified if the vectors defining the parametric system were normalised. If the point corresponding to these parameter values is (x_0', y_0', z_0'), then the implicit plane equation $ax + by + cz + d = 0$ is readily derived in its normalised form:

$$d = \sqrt{(x_0'^2 + y_0'^2 + z_0'^2)}$$

$$a = \frac{x_0'}{d}$$

$$b = \frac{y_0'}{d}$$

$$c = \frac{z_0'}{d}$$

Conversely, to go from implicit to parametric equations, we can write down (x_0, y_0, z_0) from the plane equation (assumed not to be normalised):

$$x_0 = \frac{da}{\sqrt{(a^2 + b^2 + c^2)}}$$

$$y_0 = \frac{db}{\sqrt{(a^2 + b^2 + c^2)}}$$

$$z_0 = \frac{dc}{\sqrt{(a^2 + b^2 + c^2)}}$$

Selecting the vector system for the parametric equation is somewhat arbitrary, however. We may choose a first vector for external reasons, and the second is then formed from the vector product of that vector and the normal vector (a, b, c). If we have no particular interest in the choice of the vector system, then the first vector can be created as the vector product of the normal and any arbitrary vector, which must, of course, make a reasonably large angle with it. The coordinate axis corresponding to the smallest of a, b, or c is a convenient choice.

7.8 Distance from a Point to a Plane



If the plane is $ax + by + cz + d = 0$ and the point is (x_J, y_J, z_J) then the square of the perpendicular distance from the point to the plane, r^2, is:

$$r^2 = \frac{(ax_J + by_J + cz_J + d)^2}{a^2 + b^2 + c^2}$$

This is very similar to the two-dimensional formula given in Section 1.3. The code is:

```
DENOM = A*A + B*B + C*C
IF (DENOM.LT.ACCY) THEN

          ..... The plane equation is corrupt

ELSE
          SR = A*XJ + B*YJ + C*ZJ + D
          RSQ = SR*SR/DENOM
ENDIF
```

If the plane is normalised, the single statement

```
SR = A*XJ + B*YJ + C*ZJ + D
```

suffices. As in Section 1.3, a positive value of SR indicates that the point is on the side of the plane towards which the vector (a, b, c) is pointing. Negative values indicate that it is on the other side. If this information is irrelevant the absolute value of SR should be taken.

7.9 Angle between a Line and a Plane



The angle between a line and a plane is the angle between the normal to the plane and the line, subtracted from a right angle. The first angle, between the line and the normal to the plane, may be found, as usual, from a scalar product.

If the plane is

$$ax + by + cz + d = 0$$

and the line is

$$x = x_0 + ft$$

$$y = y_0 + gt$$

$$z = z_0 + ht$$

the angle between the line and normal is

$$\gamma = \cos^{-1}(af + bg + ch)$$

if both line and plane are normalised.

The following expressions

$$\gamma = \cos^{-1}\frac{af + bg + ch}{\sqrt{(a^2 + b^2 + c^2)}}$$

$$\gamma = \cos^{-1} \frac{af + bg + ch}{\sqrt{(f^2 + g^2 + h^2)}}$$

and
$$\gamma = \cos^{-1} \frac{af + bg + ch}{\sqrt{[(a^2 + b^2 + c^2)(f^2 + g^2 + h^2)]}}$$

may be used when just the line, just the plane, and neither are normalised; they are efficient alternatives to local normalisation of the line and plane. The angle between the line and the plane is then:

$$\theta = \frac{\pi}{2} - \gamma$$

This should be in the range $-\pi/2$ to $\pi/2$. Positive values indicate that the line's parameter moves away from the plane on the same side as its normal points. Negative values indicate the reverse, and if $\theta = 0$, of course, the line is parallel to the plane.

The most general solution is coded:

```
DENOM = (A*A + B*B + C*C)*(F*F + G*G + H*H)
IF (DENOM.LT.ACCY) THEN

        ..... Either or both equations are corrupt

ELSE
        SPROD = A*F + B*G + C*H
        GAMMA = ACOS(SPROD/SQRT(DENOM))
        THETA = 1.570796 - GAMMA
ENDIF
```

7.10 Angle between Two Planes



This problem is very similar to that solved in the previous section. To find the angle between two planes we find the angle between their normals. If the two planes are

$$a_1x + b_1y + c_1z + d_1 = 0$$

and $$a_2x + b_2y + c_2z + d_2 = 0$$

then, if the coefficients are normalised, the angle between the planes is

$$\theta = \cos^{-1}(a_1a_2 + b_1b_2 + c_1c_2)$$

and with unnormalised coefficients it is:

$$\theta = \cos^{-1}\frac{a_1a_2 + b_1b_2 + c_1c_2}{\sqrt{[(a_1^2 + b_1^2 + c_1^2)(a_2^2 + b_2^2 + c_2^2)]}}$$

This latter equation will code to run faster than separately normalising the planes and then finding the angle between them. The code needed is identical to that used in the last section, and results are again in the range 0 - π. When the plane equations represent half-spaces, the angle is determined considering them in the same sense, ie the solid semi-infinite spaces would coincide if one of the half-spaces were to be rotated through θ.

7.11 Intersection of a Line and a Plane



If the plane is

$$ax + by + cz + d = 0$$

and the line is

$$x = x_0 + ft$$

$$y = y_0 + gt$$

$$z = z_0 + ht$$

then the parameter on the line where it intersects the plane is given by

$$t = \frac{-(ax_0 + by_0 + cz_0 + d)}{af + bg + ch}$$

and the intersection point can be found using the code:

```
DENOM = A*F + B*G + C*H
IF (ABS(DENOM).LT.ACCY) THEN

        ..... The line and plane are parallel

ELSE
        T = -(A*X0 + B*Y0 + C*Z0 + D)/DENOM
        X = X0 + F*T
        Y = Y0 + G*T
        Z = Z0 + H*T
```

```
ENDIF
```

This code is completely general - none of the equations need to be normalised.

7.12 Intersection of Three Planes



This problem is the three-dimensional equivalent of the two-dimensional problem of finding the point where two straight lines intersect. As before the solution is to treat the plane equations as three simultaneous linear equations and solve them. If the planes are

$$a_1x + b_1y + c_1z + d_1 = 0$$

$$a_2x + b_2y + c_2z + d_2 = 0$$

$$a_3x + b_3y + c_3z + d_3 = 0$$

then the following code will find the intersection point (x, y, z):

```
BC = B2*C3 - B3*C2
AC = A2*C3 - A3*C2
AB = A2*B3 - A3*B2
DET = A1*BC - B1*AC + C1*AB
IF (ABS(DET).LT.ACCY) THEN

        ..... At least two planes are parallel

ELSE
        DC = D2*C3 - D3*C2
        DB = D2*B3 - D3*B2
```

```
        AD = A2*D3 - A3*D2
        DETINV = 1.0/DET

        X = (B1*DC - D1*BC - C1*DB)*DETINV
        Y = (D1*AC - A1*DC - C1*AD)*DETINV
        Z = (B1*AD + A1*DB - D1*AB)*DETINV
    ENDIF
```

7.13 Intersection of Two Planes



Two planes intersect to form a straight line. If the planes are

$$a_1x + b_1y + c_1z + d_1 = 0$$

and $$a_2x + b_2y + c_2z + d_2 = 0$$

and the line of intersection that we want to find is

$$x = x_0 + ft$$

$$y = y_0 + gt$$

$$z = z_0 + ht$$

then the parameter coefficients, f, g, and h, are given by:

$$f = \begin{vmatrix} b_1 & c_1 \\ b_2 & c_2 \end{vmatrix} \qquad g = \begin{vmatrix} c_1 & a_1 \\ c_2 & a_2 \end{vmatrix} \qquad h = \begin{vmatrix} a_1 & b_1 \\ a_2 & b_2 \end{vmatrix}$$

As this stands f, g, and h are not normalised. If we choose to make the point (x_0, y_0, z_0) the point on the line of intersection nearest to the origin then the following code will find the line:

```
F = B1*C2 - B2*C1
G = C1*A2 - C2*A1
H = A1*B2 - A2*B1
DET = F*F + G*G + H*H
IF (DET.LT.ACCY) THEN

          ..... The planes are parallel

ELSE
          DETINV = 1.0/DET
          DC = D1*C2 - C1*D2
          DB = D1*B2 - B1*D2
          AD = A1*D2 - A2*D1

          X0 = (G*DC - H*DB)*DETINV
          Y0 = -(F*DC + H*AD)*DETINV
          Z0 = (F*DB + G*AD)*DETINV
ENDIF
```

If the reader wishes to normalise the line equation he should include code in the ELSE clause to multiply F, G, and H by SQRT(DETINV), but this should not work out the square root three times.

7.14 Plane through Three Points



The implicit equation of a plane through three points may be stated as a determinant containing the three independent variables x, y and z

$$\begin{vmatrix} x - x_J & y - y_J & z - z_J \\ x_K - x_J & y_K - y_J & z_K - z_J \\ x_L - x_J & y_L - y_J & z_L - z_J \end{vmatrix} = 0$$

Which effectively states that a vector formed by J and any point in the plane must be perpendicular to the vector product of the vectors from J to K and from J to L. This vector product is, of course, normal to the plane, so the variable vector from J must lie in the plane.

A normalised form of the plane equation is not usually produced, and the numerical accuracy of the code below relies on the points being spaced well apart and not all lying close to a straight line (in other words the nearer they are to forming an equilateral triangle the better).

If the determinant above is multiplied out it gives the usual form of the plane equation:

$$ax + by + cz + d = 0$$

When coding this considerable exploitation of repeated expressions can be made:

```
XKJ = XK - XJ
YKJ = YK - YJ
ZKJ = ZK - ZJ
XLJ = XL - XJ
YLJ = YL - YJ
ZLJ = ZL - ZJ

A = YKJ*ZLJ - ZKJ*YLJ
B = ZKJ*XLJ - XKJ*ZLJ
C = XKJ*YLJ - YKJ*XLJ
D = -( XK*A + YK*B + ZK*C )
```

7.15 Plane through a Point and Normal to a Line



This is not a difficult calculation, because a plane is described in terms of its normal. If the line is

$$x = x_0 + ft$$

$$y = y_0 + gt$$

$$z = z_0 + ht$$

then the plane normal to that line passing through a point J is

$$fx + gy + hz - (fx_J + gy_J + hz_J) = 0$$

The plane will only be normalised if the line was.

7.16 Plane through Two Points and Parallel to a Line



If the given line is

$$x = x_0 + ft$$

$$y = y_0 + gt$$

$$z = z_0 + ht$$

then the plane parallel to it which passes through K and L is:

$$(hy_{LK} - gz_{LK})x + (fz_{LK} - hx_{LK})y + (gx_{LK} - fy_{LK})z -$$
$$x_K(hy_{LK} - gz_{LK}) - y_K(fz_{LK} - hx_{LK}) - z_K(gx_{LK} - fy_{LK}) = 0$$

where $x_{LK} = x_L - x_K$, $y_{LK} = y_L - y_K$, and $z_{LK} = z_L - z_K$

This, despite its complexity, is not even normalised. However, the number of repeated terms means that the code is relatively simple, and that it is also easy to produce a normalised answer. If we consider the above equation to reduce to

$$ax + by + cz + d = 0$$

then to get the normalised coefficient in this equation we may use the code:

```
XLK = XL - XK
YLK = YL - YK
ZLK = ZL - ZK
A = H*YLK - G*ZLK
B = F*ZLK - H*XLK
C = G*XLK - F*YLK
DENOM = A*A + B*B + C*C
IF (DENOM.LT.ACCY) THEN

        ..... Points coincident or line coefficients corrupt

ELSE
        DENINV = 1.0/SQRT(DENOM)
        A = A*DENINV
        B = B*DENINV
        C = C*DENINV
        D = -(A*XK + B*YK + C*ZK)
ENDIF
```

Note that the point (x_0, y_0, z_0) is not needed to solve this problem.